

ALLSEL	!选中所有对象
LSEL,S,LINE,,6	!选中 6 号线
LSEL,A,LINE,,8	!继续选中 8 号线
LSEL,A,LINE,,10	!继续选中 10 号线
LATT,1,3,1	!为线分配材料类型、实常数和单元类型
ALLSEL	!选中所有对象
LSEL,S,LINE,,11	!选中 11 号线
LSEL,A,LINE,,13	!继续选中 13 号线
LSEL,A,LINE,,15,16,1	!继续选中 15、16 号线
LATT,1,5,2	!为线分配材料类型、实常数和单元类型
ALLSEL	!选中所有对象
LSEL,S,LINE,,12	!选中 12 号线
LSEL,A,LINE,,14	!继续选中 14 号线
LATT,1,2,1	!为线分配材料类型、实常数和单元类型
ALLSEL	!选中所有对象
LMESH,1,16,1	!划分 1~16 号线单元
NSEL,S,LOC,Z,15	!选中 Z=15 处的所有节点
CP,,ALL,ALL	!耦合所选节点的全部自由度
ALLSEL	!选中所有对象
<b>!*****静力求解计算</b>	
/SOLU	!进入求解计算模块
ANTYPE,0	!设置求解类型为结构静力计算
NSEL,S,LOC,Z,-37	!选中 Z=-37 处的所有节点
D,ALL,ALL	!约束所选节点的全部自由度
ALLSEL	!重新选中所有对象
NSEL,S,LOC,X,11.5	!选中 X=11.5 处的所有节点
NSEL,R,LOC,Z,-35	!从已选节点中重新选中 Z=-35 处的所有节点
D,ALL,ALL	!约束所选节点的全部自由度
ALLSEL	!重新选中所有对象
NSEL,S,LOC,Z,15	!选中 Z=15 处的所有节点
NSEL,R,LOC,X,0	!从已选节点中重新选中 X=0 处的所有节点
F,ALL,FZ,-2E6	!对所选节点施 Z 向的载荷
ALLSEL	!重新选中所有对象
NSEL,S,LOC,Z,1,15	!选中 Z=1~15 范围内的所有节点
NSEL,R,LOC,X,0	!从已选节点中重新选中 X=0 处的所有节点
F,ALL,FX,17829/15	!对已选节点施加 X 方向的载荷

```

ALLSEL          !重现选中全部对象
ACEL,0,0,9.8    !施加重力载荷
SOLVE           !执行求解计算
SAVE            !保存求解结果
FINI            !退出求解计算模块
!*****静力分析后处理
/POST1          !进入通用后处理器
SET, LAST       !读入计算结果文件
/ESHAPE, 1.0    !打开模型线形状选项
PLDISP, 0       !绘制结构整体变形图
PLNSOL, U, SUM, 0, 1.0 !绘制结构位移等值图
PLNSOL, S, EQV, 0, 1.0 !绘制结构等效应力分布云图
ESEL, S, REAL, , 3 !选中内筒单元
PLNSOL, S, EQV, 0, 1.0 !绘制内筒等效应力分布云图
FINISH          !退出后处理器
!*****结构模态分析
/PREP7          !重新进入前处理模块
ET, 3, MASS21   !定义 3 号单元类型 MASS21
R, 7, , , 2E5   !定义 7 号实常数（针对 MASS21 单元）
TYPE, 3         !选中 3 号单元类型
REAL, 7         !选中 7 号实常数
NSEL, S, LOC, Z, 15 !选中 Z=15 处的节点
NSEL, R, LOC, X, 0 !在已选中的节点中重新选取 X=0 处的节点
E, ALL          !通过已选中的节点创建 MASS21 单元
ALLSEL          !选中所有对象
NSEL, S, LOC, Z, -36, 15 !选中 Z=-36~15 范围内的所有节点
NSEL, R, LOC, X, 0, 0.290 !在已选中的节点中重新选取 X=0~0.290 范围内的节点
CPINTF, ALL, 0.8 !将所选中的节点进行耦合
ALLSEL          !选中所有对象
NSEL, S, LOC, Z, -36, 15 !选中 Z=-36~15 范围内的所有节点
NSEL, R, LOC, X, 0, 0.290 !在已选中的节点中重新选取 X=0~0.290 范围内的节点
D, ALL, UY      !约束所选中节点的 Y 方向位移自由度
D, ALL, UZ      !约束所选中节点的 Z 方向位移自由度
D, ALL, ROTX    !约束所选中节点的 X 方向转动自由度
D, ALL, ROTZ    !约束所选中节点的 Z 方向转动自由度
ALLSEL          !选中所有对象

```

```

NSEL,S,LOC,Z,-37      !选中 Z=-37 处的节点
D,ALL,ALL              !约束所选中节点的全部自由度
ALLSEL                !选中所有对象
NSEL,S,LOC,X,11.5     !选中 X=11.5 处的节点
D,ALL,ALL              !约束所选中节点的全部自由度
ALLSEL                !选中所有对象
ACEL,0,0,9.8          !施加重力载荷
/SOLU                  !进入求解计算模块
ANTYPE,MODAL           !设置分析类型为模态分析
MODEOPT,LANB,2        !模态求解方法设置
MXPAND,2              !扩展两阶模态
SOLVE                  !执行模态求解计算
FINISH                 !退出求解计算模块
/POST1                 !进入通用后处理器
SET,LIST               !列表显示求解结果
SET,FIRST              !读取一阶模态结果
PLDISP,0              !绘制一阶模态图形
SET,NEXT               !读取二阶模态结果
PLDISP,0              !绘制二阶模态图形
FINISH                 !退出后处理模块
!*****瞬态动力分析
/PREP7                !进入前处理模块
TB,81,0                !修改 WATERTABLE 里的波浪相位角为 0°
/SOLU                  !进入求解计算模块
ANTYPE,TRANS           !设置求解类型为瞬态动力分析
ALPHAD,0.3331          !设置阻尼系数  $\alpha$ 
BETAD,1.834E-3         !设置阻尼系数  $\beta$ 
TRNOPT,FULL            !设定瞬态动力求解方法
NLGEOM,ON              !打开大变形开关
NSUBST,1,,,1
*DO,I,10E-12,100,0.2   !设定结束时间及子步时间开始循环计算
TIME,I                 !循环计算结束时间
OUTPES,ALL,ALL         !输出结果设置
SOLVE,SAVE             !执行求解并保存
*ENDDO                 !结束循环求解计算

```

**!\*\*\*\*\*时程后处理过程**

/POST26	!进入时程后处理器 POST26
NSOL,2,1,U,X,Deflection	!定义 1 号节点处的 X 方向位移为 2 号时程变量
/AXLAB,X,Time	!定义 X 轴标题
/AXLAB,Y,Deflection	!定义 Y 轴标题
PLVAR,2	!绘制时程位移曲线
ESOL,3,17, ,SMIS,15,STRESS	!定义 17 号单元的等效应力为 3 号时程变量
/AXLAB,X,Time	!定义 X 轴标题
/AXLAB,Y,Stress	!定义 Y 轴标题
PRVAR,3	!绘制时程应力曲线
FINISH	!退出 POST26

## 第8章 深水隔水导管与导向孔相互作用分析

### ◇ 本章导读

对于水深较深的海域，需要安装导向孔来固定隔水导管，使得其能够满足作业要求，本章将对二者之间相互作用的力学问题进行分析，结合工程实例介绍 ANSYS 解决此类问题的一些基本方法。

本章包括如下的一些问题：

- 隔水导管导向孔简述
- 工程实例：某油田隔水导管与导向孔相互作用研究

### 8.1 隔水导管导向孔简述

海上钻井隔水导管是从海上钻井平台下到海底浅层的套管，主要功能是隔离海水、形成钻井液循环通道，同时作为海上井口的持力结构。在海上石油的勘探中，钻井隔水导管的稳定性对于整个石油的勘探、开采起着非常重要的作用。对于水深较深的海域，海浪流会对隔水导管的稳定性产生很大的影响。为了能够在恶劣环境载荷下正常作业不至于导管结构失效，一般情况下需要沿着隔水导管长度方向在水面与水下一定位置安装导向孔，各层导向孔槽口使用扶正块来补偿隔水导管和导管架导向孔槽口之间的间隙，以减少隔水导管承受的疲劳应力和涡激应力，延长隔水导管的使用寿命。一般情况下，先将导向孔固定在导管架平台的桩腿上，再使用扶正块来补偿隔水导管与导向孔之间的间隙，用以固定隔水导管，从结构上来讲，就是减小隔水导管的自由长度降低有效长细比来提高其稳定性。

但是随着水深的增加，水下导向孔及扶正块的安装越来越困难，目前国内公司可以实现扶正块安装作业的水深能力有限。所以进行对隔水导管导向孔安装位置及个数的优化研究显得尤为重要，研究结果能够更好地为海上钻井隔水导管导向孔的施工提供科学依据和安全保障，进一步减少盲目性和复杂事故的发生。

钻井隔水导管受到的主要载荷包括：上端部井口载荷，结构自重，浮力等轴向载荷；另外主要受到海洋风浪流等复杂环境载荷。对于隔水导管下部边界条件的处理方法是根据有效桩长法，将导管泥面以下六倍桩径处看作固支端，上部边界横向位移受铰支约束，轴向可以自由滑动。对于导向孔采用以下假设处理：

1) 导向孔固定于导管架上, 看作为固定构件, 为隔水导管提供稳定的横向支撑;

2) 导向孔与隔水导管之间间隙用扶正块将其固定, 隔水导管横向位移自由度受到约束, 因此, 在分析中可将安装导向孔位置处的导管模型约束其横向位移自由度。隔水导管力学分析模型如图 8-1 所示。




图 8-1 隔水导管力学分析模型

## 8.2 工程实例：某油田隔水导管与导向孔相互作用研究

以某油田为例, 隔水导管导向孔安装位置如图 8-3 所示。该油田所在的海域水深大约为 117m。原来设计在水面以下安装四层导向孔, 水面以上安装一层导向孔, 具体的位置在钻井转盘面以下 30.10m (水上), 52.60m (水下第一层), 76.60m (水下第二层), 101.60m (水下第三层), 128.60m (水下第四层)。




图 8-3 导向孔安装示意图

可以看出对于水下第四层导向孔及扶正块的安装难度极大，超过了目前国内公司的作业能力。因此需要对不同海况条件下的不同导向孔个数进行力学分析，分析结果可以为是否有必要安装第四层导向孔提供可靠的理论依据。其具体参数如下：

- 隔水导管尺寸

外径：508mm， 壁厚：25.4mm。

- 材料参数

弹性模量：2.1E11Pa， 泊松比：0.3， 密度：7850kg/m<sup>3</sup>。

- 环境参数：海况重现期 50 年的环境载荷条件为：

风速 45.2m/s，有效波高 12.8m，有效周期 13.1s，

海流表层流速 2.21m/s，中部 1.32m/s，底部 0.74m/s。

水深：117.63m，海面以上高度：38m，

- 顶部载荷：30T。

## 8.2.1 模型建立

### (1) 设置工作环境

进入 ANSYS/Multiphysics 的程序界面后，指定分析的工作名称为“Drilling riser optimization”，指定图形显示区域的标题为【Analysis of Drilling riser optimization】。

### (2) 定义单元类型

选择菜单项【Main Menu】>【Preprocessor】进入前处理器 PREP7 开始建模以及其他的前处理操作。单击菜单【Main Menu】>【Preprocessor】>【Element Type】>【Add/Edit/Delete】，定义两种单元类型“PIPE16”和“PIPE59”。

ET,1,PIPE16	!定义 1 号单元类型 PIPE16
ET,2,PIPE59	!定义 2 号单元类型 PIPE59

### (3) 设置单元实常数

点击菜单项【Main Menu】>【Preprocessor】>【Real Constants】>【Add/Edit/Delete】，分别设置 PIPE16 单元的外径为和壁厚，PIPE59 单元的外径、壁厚及相关的流体参数设置。具体如下面的命令流：

R,1,0.508,0.0254	!定义 1 号实常数（针对 PIPE16 单元）
R,2,0.508,0.0254, 0.7,2.0,1030,0,	!定义 2 号实常数（针对 PIPE59 单元）

#### (4) 设置材料模型参数及塑性选项

选择菜单路径【Main Menu】>【Preprocessor】>【Material Props】>【Material Models】,将出现“Define Material Model Behavior”对话框,在窗口的右侧,依次单击【Structural】>【Linear】>【Elastic】>【Isotropic】,在出现的对话框中输入材料弹性模量“2.1E11”以及泊松比“0.3”,如图所示 4-8 所示,单击 OK 按钮,继续返回点击【Density】弹出密度设置窗口,设置密度为“7850”,单击【Fluids】>【Water Table】设置环境载荷参数,具体见本章附录命令流所示。

#### (5) 建立几何模型

本例将采用直接通过建立节点的方法建立有限元模型,由于在使用 PIPE59 单元时,坐标原点需设置在水面处。因此在建模时,将对图 8-3 所示的坐标进行相应的修改。

##### 1) 建立单元节点

选择菜单路径【Main Menu】>【Preprocessor】>【Modeling】>【Create】>【Nodes】>【In Active Cs】,建立位于转盘面处导管的节点,编号为“1”,坐标为 (0, 0, 38),单击【Apply】按钮,继续定义水上导向孔位置的节点,编号“31”,坐标为 (0, 0, 7.9),接着水面处节点,编号为“39”,坐标为 (0, 0, 0),依次定义水下第一层导向孔处节点,编号“54”,坐标为 (0, 0, -14.6),水下第二层导向孔处节点,编号“78”,坐标为 (0, 0, -38.6),水下第三层导向孔处节点,编号“103”,坐标为 (0, 0, -63.6),水下第四层导向孔处节点,编号“130”,坐标为 (0, 0, -90.6),水下井口处(即泥线处)节点,编号“157”,坐标为 (0, 0, -117.63),导管固支端节点,编号“160”,坐标为 (0, 0, -120.63),单击【OK】按钮。

选择菜单【Main Menu】>【Preprocessor】>【Modeling】>【Create】>【Nodes】>【Fill between Nds】,弹出节点拾取对话框,利用鼠标在图形显示区域拾取节点“1”和“31”,单击【OK】按钮,在弹出的对话框中默认所有选项,单击【Apply】按钮进行填充操作,按照相同方法填充其他节点,结果如图 8-4 所示。此部分的命令流操作如下:

N,1,0,0,38	!建立 1 号节点
N,31,0,0,7.9	!建立 31 号节点
FILL	!在 1 号和 31 号节点之间生成 2~30 号节点
N,39,0,0,0	!建立 39 号节点
FILL	!在 31 号和 39 号节点之间生成 2~38 号节点
N,54,0,0,-14.6	!建立 54 号节点
FILL	!在 39 号和 54 号节点之间生成 40~53 号节点
N,78,0,0,-38.6	!建立 78 号节点
FILL	!在 54 号和 78 号节点之间生成 55~77 号节点
N,103,0,0,-63.6	!建立 103 号节点
FILL	!在 78 号和 103 号节点之间生成 79~102 号节点



N,130,0,0,-90.6	!建立 130 号节点
FILL	!在 103 号和 130 号节点之间生成 104~129 号节点
N,157,0,0,-117.63	!建立 157 号节点
FILL	!在 130 号和 157 号节点之间生成 131~156 号节点
N,160,0,0,-120.63	!建立 160 号节点
FILL	!在 157 号和 160 号节点之间生成 158、159 号节点




图 8-4 节点建立结果

## 2) 通过节点建立单元

选择菜单路径【Main Menu】>【Preprocessor】>【Modeling】>【Create】>【Elements】>【Elem Attributes】，弹出“Element Attributes”对话框，在【TYPE】选项后的下拉框中选中“2 PIPE59”，在【MAT】选项后的下拉列表中选中“1”，在【REAL】选项后的下拉列表中选择“2”单元实常数。单击【OK】按钮。单击【Main Menu】>【Preprocessor】>【Modeling】>【Create】>【Elements】>【Auto Numbered】>【Thru Nodes】，弹出节点拾取对话框，利用鼠标在图形显示区域选中节点 1 和 2，单击【Apply】按钮，继续选择节点 2 和 3，3 和 4，...，43 和 44。单击【OK】按钮。

按照上述相同的操作过程，创建泥线以下桩腿单元和非线性弹簧单元，最终建立的有限元模型如图 5-10 所示。如采用循环语句建立单元的命令流如下：

TYPE,2	!选中 2 号单元类型
REAL,2	!选中 2 号单元类型
MAT,1	!选中 1 号材料模型
*DO,I,1,156,1	!参数化建模

E,I,I+1	!通过相邻节点建立单元模型
*ENDDO	!完成泥线以上 PIPE59 单元模型
TYPE,1	!选中 1 号单元类型
REAL,1	!选中 1 号单元类型
MAT,1	!选中 1 号材料模型
*DO,I,157,159,1	!参数化建模
E,I,I+1	!通过相邻节点建立单元模型
*ENDDO	!完成泥线以下 PIPE16 单元

### 3) 导向孔处理

根据模型假设，将安装导向孔位置处的导管模型按约束其横向位移自由度来处理。

选择常用工具栏菜单【Select】>【Entities】，弹出选择项的属性设置窗口，在第一个下拉菜单路径中选中“Nodes”，第二个下拉菜单路径中选中“By Location”并选中“Z coordinates”，在“Min, Max”方框中输“7.9”，并选中【From All】，单击【OK】按钮。继续选择【Select】>【Entities】，选中“Nodes”>“By Location”>“Z coordinates”，在“Min, Max”方框中输“-14.6”，并选中【Also Select】，单击【OK】，重复操作选中位于所有导向孔的节点。

选择菜单路径项【Main Menu】>【Preprocessor】>【Loads】>【Define Loads】>【Apply】>【Structural】>【Displacement】>【On Nodes】，弹出节点拾取窗口，选中【Pick All】按钮，弹出自由度设置窗口，选择“UX,UY”，单击【OK】退出。选择常用工具栏菜单【Select】>【Everything】。命令流如下：

NSEL,S,LOC,Z,7.9	!选中 Z=7.9 处的节点
NSEL,A,LOC,Z,-14.6	!继续选中 Z=-14.6 处的节点
NSEL,A,LOC,Z,-38.6	!继续选中 Z=-38.6 处的节点
NSEL,A,LOC,Z,-63.6	!继续选中 Z=-63.6 处的节点
NSEL,A,LOC,Z,-90.6	!继续选中 Z=-90.6 处的节点（选择性约束）
D,ALL,UX	!约束所选节点 X 方向位移自由度
D,ALL,UY	!约束所选节点 Y 方向位移自由度
ALLSEL	!选中所有对象

### (6) 保存模型退出前处理器

选择菜单路径项【Main Menu】>【Finish】，单击常用工具条上的【SAVE DB】按钮，保存有限元模型，退出前处理模块。

## 8.2.2 静力分析

在计算部分，分别对水下四层导向孔和三层导向孔两种模型分别计算。

### (1) 设置分析类型

选择菜单项【Main Menu】>【Solution】>【Analysis Type】>【New Analysis】，弹出如图 6-15 所示分析类型选择窗口，选择“Static”，点击【OK】按钮退出。

### (2) 定义边界条件

选择菜单项【Main Menu】>【Solution】>【Define Loads】>【Apply】>【Structural】>【Displacement】>【On Nodes】，弹出节点拾取窗口，利用鼠标在图形显示区域选中导管底端节点，单击【OK】按钮，弹出自由度设置对话框，选择“All DOFs”，单击【OK】按钮，继续选中顶端节点约束其“UX,UY”方向自由度。命令流如下：

NSEL,S,LOC,Z,-120.63	!选中 Z=-120.63 处的节点
D,ALL,ALL	!约束所选节点 X 方向位移自由度
ALLSEL	!选中所有对象
NSEL,S,LOC,Z,38	!选中 Z=38 处的节点
D,ALL,UX	!约束所选节点 X 方向位移自由度
D,ALL,UY	!约束所选节点 Y 方向位移自由度
ALLSEL	!选中所有对象

### (3) 施加载荷

#### 1) 定义导管顶部载荷：

单击菜单【Main Menu】>【Solution】>【Define Loads】>【Apply】>【Structural】>【Force/Moment】>【On Nodes】，弹出节点拾取对话框，用鼠标在图形显示区域拾取导管顶部节点，单击【OK】按钮，弹出 Apply F/M on Nodes 属性值设置对话框，在“Direction of force/moment”中选择“Fz”，在 Force/moment value 中输入“-3E5”，单击 OK 按钮。

#### 2) 定义风载荷：

选择菜单路径【Main Menu】>【Solution】>【Define Loads】>【Apply】>【Structural】>【Force/Moment】>【On Nodes】，弹出“Apply F/M on Nodes”对象拾取对话框，选择【Box】，用鼠标在图形显示区域拾取水面以上所有节点，单击【OK】按钮，弹出“Apply F/M on Nodes”属性值设置对话框，在【Direction of force/moment】中选择【FX】，在【Force/moment value】中输入“18987/38”（共 38 个节点），单击【OK】按钮退出。

#### 3) 定义重力场：

单击菜单【Main Menu】>【Solution】>【Define Loads】>【Apply】>【Structural】>【Inertia】>【Gravity】>【Global】，弹出“Apply (Gravitational) Acceleration”属性值设置对话框，在“ACELX,

ACELY, ACELZ”中依次输入“0, 0, 9.8”。

#### (4) 求解

通过菜单项【Main Menu】>【Solution】>【Solve】>【Current LS】，对问题进行求解。在求解过程中会看到如图所示的求解收敛曲线。在求解结束后，弹出 Solution is done! 信息提示框，关闭。单击【Main Menu】>【Finish】菜单项，退出求解计算模块。

#### (5) 静力结果分析

##### 1) 读入结果文件

选择菜单路径【Main Menu】>【General Postproc】>【Read Results】，单击【Last Set】，读入最后一个子步的结果文件。

##### 2) 绘制结构整体变形图

选择菜单路径【Main Menu】>【General Postproc】>【Plot Results】>【Deformed Shape】，在弹出的对话框中选择【Def shape only】，显示结构整体变形图，如图 8-5 和 8-6 所示，其中图 8-5 为水下四层导向孔，图 8-6 为水下三层导向孔。



图 8-5 结构整体变形图（水下四层导向孔）

##### 3) 绘制结构节点位移等值线云图

选择菜单路径【Main Menu】>【General Postproc】>【Plot Results】>【Contour Plot】>【Nodal Solu】，弹出“Contour Nodal Solution Data”对话框，依次单击【Nodal Solution】>【DOF Solution】>【Displacement vector sum】，单击【OK】按钮，显示如图 8-7 与图 8-8 所示的结构节点位移等值线云图。从图中可以看出，水下四层导向孔结构发生最大位移的位于水面以下 5m 左右，数值为

0.0262m；水下三层导向孔结构发生最大位移的位于水面以下 93m 左右，即原来第四层导向孔位置附近，数值为 0.0608m。由此可以看出，如果去掉位于水下 90.36m 处的导向孔，整个结构的横向位移极值为 6.08cm。



图 8-6 结构整体变形图（水下三层导向孔）




图 8-7 节点位移等值线云图（水下四层导向孔）




图 8-8 节点位移等值线云图（水下三层导向孔）

## 4) 绘制单元应力等值线云图

选择菜单路径【Main Menu】>【General Postproc】>【Plot Results】>【Contour Plot】>【Element Solu】，弹出“Contour Element Solution Data”对话框，依次单击【Element Solution】>【Stress】>【von Mises stress】，单击【OK】按钮，如图 8-9 和图 8-10 所示分别为水下四层和三层导向孔结构的米塞斯等效应力云图。




图 8-9 单元等效应力等值线云图（水下四层导向孔）




图 8-10 单元等效应力等值线云图（水下三层导向孔）

由图可以看出，水下四层导向孔和三层导向孔结构的最大米塞斯应力都位于水下第一层导向孔位置附近，前者最大值为 50.3MPa，后者为 51.3MPa，由此可见，两者相差很小，但从强度角度来看，完全可以去掉水下第四层导向孔。

#### 4) 定义单元表项

通过单元输出项所规定的规则（可以在 ANSYS 帮助系统中查看对应单元类型的输出规则）定义需要查看的结果项，可以提取数据进行分析。下面进行单元等效应力及弯矩的定义。

选择菜单路径【Main Menu】>【General Postproc】>【Element Table】>【Define Table】，弹出“Element Table Data”对话框，单击【Add】按钮，在弹出窗口的【Lab】选项后的文本框中输入“MOMENT\_I”，定义单元 I 节点的弯矩，在【Item, Comp】选项后的左下拉框选中“By sequence num”，然后在右边的下拉列表框中选择“SMISC,”，接着在其下面的文本框中将内容补充为“SMISC,5”，单击【Apply】按钮，继续在【Lab】选项后的文本框中输入“MOMENT\_J”，定义单元 J 节点的弯矩，在【Item, Comp】选项后的左下拉框选中“By sequence num”，然后在右边的下拉列表框中选择“SMISC,”，接着在其下面的文本框中将内容补充为“SMISC,11”，单击【Apply】按钮。

继续定义单元的等效应力，弹出窗口的【Lab】选项后的文本框中输入“STRESS\_I”，定义单元 I 节点等效应力，在【Item, Comp】选项后的左下拉框选中“By sequence num”，然后在右边的下拉列表框中选择“NMISC,”，接着在其下面的文本框中将内容补充为“NMISC,15”，单击【Apply】按钮，继续在【Lab】选项后的文本框中输入“STRESS\_J”，定义单元 J 节点的等效应力，在【Item, Comp】选项后的左下拉框选中“By sequence num”，然后在右边的下拉列表框中选择“NMISC,”，接着在其下面的文本框中将内容补充为“NMISC,55”，单击【OK】按钮退出。

选择菜单路径【Main Menu】>【General Postproc】>【Plot Results】>【Contour Plot】>【Line Element Res】，弹出“Plot Line-Element Results”对话框，在【LabI】选项后的下拉列表中选中“MOMENT\_I”，在【LabJ】选项后的下拉列表中选中“MOMENT\_J”，其他选项默认，单击【OK】按钮，显示出如图 8-11 和图 8-12 所示的单元弯矩分布云图。由图中可以看出，结构最大弯矩出现在水面附近。




图 8-11 结构弯矩分布图（水下四层导向孔）




图 8-12 结构弯矩分布图（水下三层导向孔）

从图 8-11 和图 8-12 中可以看出，两种结构的弯矩分布大致相同，其最大值都分布在水面处附近，水下三层导向孔结构的极限弯矩相比水下四层导向孔结构来说有所减小。分别为



142988N\*m 和 144325N\*m。

选择菜单路径【Main Menu】>【General Postproc】>【Plot Results】>【Contour Plot】>【Line Element Res】，弹出“Plot Line-Element Results”对话框，在【LabI】选项后的下拉列表中选中“STRESS\_I”，在【LabJ】选项后的下拉列表中选中“STRESS\_J”，其他选项默认，单击【OK】按钮，显示出如图 8-13 和图 8-14 所示的单元应力分布云图。由图中可以看出，结构最大弯矩出现在水面附近。




图 8-13 结构应力分布图（水下四层导向孔）




图 8-14 结构应力分布图（水下三层导向孔）

##### 5) 列表显示所定义的单元表项

选择菜单路径【Main Menu】>【General Postproc】>【Element Table】>【List Elem Table】，弹出如图 8-15 所示的对话框，在右侧下拉列表框中选中“MOMENT\_I, MOMENT\_J, STRESS\_I, STRESS\_J”，单击【OK】按钮。




图 8-15 单元表项设置

## 8.2.3 结构模态分析

### (1) 模型修改

打开 ANSYS 程序主界面，导入静力分析时所建立的的平台桩腿建模部分，选择菜单路径【Main Menu】>【Preprocessor】>【Element Type】>【Add/Edit/Delete】，弹出【Element Type】对话框，单击【Add】按钮，定义 MASS21 为 3 号单元类型，定义实常数“R,3,,,3E4”。

选择菜单路径【Main Menu】>【Preprocessor】>【Modeling】>【Create】>【Elements】>【Elem Attributes】，弹出单元属性设置对话框，选择单元类型为“3 MASS21”，实常数为“24”，其他默认，单击【OK】退出。选择【Main Menu】>【Preprocessor】>【Modeling】>【Create】>【Elements】>【Auto Numbered】>【Thru Nodes】，弹出的节点拾取对话框，在图形显示区域中，用鼠标选中桩腿最顶端的节点，单击【OK】按钮，集中质量单元创建完成，模型修改完成。

### (2) 设置位移约束

进行结构模态分析时，为了略去结构无关的振型，对导管结构除导向孔位置以外的所有节点只保留 X 和 ROTY 两个自由度。

### (3) 设定模态分析类型

重新进入 ANSYS 求解器，选择菜单路径【Main Menu】>【Solution】>【Analysis Type】>【New Analysis】，设置分析类型为【Modal】。

### (4) 设置分析选项

选择菜单路径【Main Menu】>【Solution】>【Analysis Type】>【Analysis Option】，弹出“Modal Analysis”对话框，模态提取方法采用【Block Lanczos】，提取模态数设为“6”。

### (5) 扩展模态设置

选择菜单路径【Main Menu】>【Solution】>【Load Step Opts】>【ExpansionPass】>【Single Expand】>【Expand Modes】，弹出“Expand Nodes”对话框，模态扩展数设为“6”。

### (6) 模态分析求解

选择菜单路径【Main Menu】>【Solution】>【Solve】>【Current LS】，弹出“Solve Current Load Step”对话框，单击【OK】，开始计算模态解。求解完毕后，在“Note”窗口显示“Solution is done! ”，单击【Close】关闭窗口。

### (5) 观察模态分析结果

#### 1) 观察模型固有频率

进入通用后处理器 POST1，选择菜单路径【Main Menu】>【General Postproc】>【Results Summary】，观察模型的固有频率，结果如表 8.1 所示。

表 8.1 结构固有频率

振型		一阶	二阶	三阶	四阶	五阶	六阶
频率 (Hz)	水下四层	0.3133	0.3964	1.7454	2.1513	2.4946	2.8426
	水下三层	0.0988	0.3963	0.6498	1.7677	2.1662	2.5886

#### 2) 读入结果文件

选择菜单路径【Main Menu】>【General Postproc】>【Read Results】>【First Set】，读入第 1 载荷子步的计算结果。

#### 3) 观察模态振型

选择菜单路径【Main Menu】>【General Postproc】>【Plot Results】>【Deformed Shape】，显示一阶模态振型，如图 8-16 所示。

为了便于观察，建议选择菜单路径【Utility Menu】>【PlotCtrls】>【Style】>【Size and Shape】，在“Size and Shape”对话框打开“Display of element shapes based on real constant descriptions”选项，以便在图形窗口以单元实际截面形状显示模型的模态振型。

重复上述操作可以得出其他五阶模态振型，如图 8-17 至 8-21 所示，图 8-22-至 8-27 所示为水下三层导向孔前六阶模态图。




图 8-16 一阶模态（水下四层导向孔）

图 8-17 二阶模态（水下四层导向孔）




图 8-18 三阶模态（水下四层导向孔）

图 8-19 四阶模态（水下四层导向孔）




图 8-20 五阶模态（水下四层导向孔）

图 8-21 六阶模态（水下四层导向孔）




图 8-22 一阶模态 （水下三层导向孔）




图 8-23 二阶模态 （水下三层导向孔）




图 8-24 三阶模态 （水下三层导向孔）




图 8-25 四阶模态 （水下三层导向孔）




图 8-25 五阶模态 （水下三层导向孔）




图 8-27 六阶模态 （水下三层导向孔）

由图可知，水下四层和三层导向孔两种结构的前两阶振型相同，二者自振频率相差很小。因此，从计算结果可以看出，去掉水下第四层导向孔对隔水导管的整体结构强度和性能影响不大，是可以满足工程需要的。

本章主要介绍了深水隔水导管与导向孔之间的相互作用模拟分析。首先隔水导管及其导向孔的简要介绍，接着介绍二者之间相互作用的力学分析模拟方法及在 ANSYS 中的实现。通过工程实例详细介绍了隔水导管与导向孔在进行力学分析时的具体操作过程，并对工程实例进行了不同工况条件下的优化计算，为实际工程施工提供了理论依据。通过本章的讲解，读者可以掌握类似结构的的处理方法和计算过程。

\*\*\*\*\*

```

某油田隔水导管与导向孔相互作用分析命令流
!*****
/FILNAME,Drilling riser optimization,!设置工作文件名
/TITLE,Analysis of Drilling riser optimization !设置图形显示区域名
!*****定义环境载荷参数
Dwater=117.63 !水深
H_wave=12.8 !波高
T_wave=13.1 !波周期
V_surface=2.21 !水面处海流速度
V_middle=1.32 !中部海流速度
V_bottom=0.74 !底部海流速度
! *****定义导管参数
OD=0.508 !导管外径
TW=0.0254 !导管壁厚
!*****定义桩顶载及风载
F_top=-3e5 !井口压载
F_wind=18978 !风载荷
!*****前处理过程
/PREP7 !进入前处理
ET,1,PIPE16 !定义 1 号单元类型 PIPE16
ET,2,PIPE59 !定义 2 号单元类型 PIPE59
R,1,0.508,0.0254 !定义 1 号实常数（针对 PIPE16 单元）
R,2,0.508,0.0254,0.7,2.0,1030,0, !定义 2 号实常数（针对 PIPE59 单元）
MP,EX,1,2.1E11 !设置弹性模量

```

```

MP,PRXY,1,0.3      !设置泊松比
MP,DENS,1,7850      !设置密度
TBDE,WATE,1         !设置波流参数 Water Table
TB,WATE,1,,,0       !Water Table
TBMODIF,1,1,2       !选择 Stokes 五阶波理论
TBMODIF,1,2,0       !波流耦合作用方式
TBMODIF,1,3,Dwater  !定义水深
TBMODIF,1,4,1030    !海水密度
TBMODIF,1,5,0       !波浪作用方向
TBMODIF,2,1,-Dwater !底部坐标
TBMODIF,2,2,V_bottom !底部流速
TBMODIF,2,3,0       !底部海流作用方向
TBMODIF,2,4,- Dwater/2 !中部坐标
TBMODIF,2,5,V_middle !中部流速
TBMODIF,2,6,0       !中部海流作用方向
TBMODIF,3,1,0       !水面坐标
TBMODIF,3,2,V_surface !海面流速
TBMODIF,3,3,0       !表面海流作用方向
TBMODIF,14,1,H_wave !波高
TBMODIF,14,2,T_wave !波周期
TBMODIF,14,3,43     !波浪相位角
N,1,0,0,38          !建立 1 号节点
N,31,0,0,7.9        !建立 31 号节点
FILL                 !在 1 号和 31 号节点之间生成 2~30 号节点
N,39,0,0,0          !建立 39 号节点
FILL                 !在 31 号和 39 号节点之间生成 2~38 号节点
N,54,0,0,-14.6      !建立 54 号节点
FILL                 !在 39 号和 54 号节点之间生成 40~53 号节点
N,78,0,0,-38.6      !建立 78 号节点
FILL                 !在 54 号和 78 号节点之间生成 55~77 号节点
N,103,0,0,-63.6     !建立 103 号节点
FILL                 !在 78 号和 103 号节点之间生成 79~102 号节点
N,130,0,0,-90.6     !建立 130 号节点
FILL                 !在 103 号和 130 号节点之间生成 104~129 号节点
N,157,0,0,-117.63   !建立 157 号节点
FILL                 !在 130 号和 157 号节点之间生成 131~156 号节点

```

```

N,160,0,0,-120.63      !建立 160 号节点
FILL                    !在 157 号和 160 号节点之间生成 158、159 号节点
TYPE,2                  !选中 2 号单元类型
REAL,2                  !选中 2 号单元类型
MAT,1                   !选中 1 号材料模型
*DO,I,1,156,1           !参数化建模
E,I,I+1                 !通过相邻节点建立单元模型
*ENDDO                  !完成泥线以上 PIPE59 单元模型
TYPE,1                  !选中 1 号单元类型
REAL,1                  !选中 1 号单元类型
MAT,1                   !选中 1 号材料模型
*DO,I,157,159,1         !参数化建模
E,I,I+1                 !通过相邻节点建立单元模型
*ENDDO                  !完成泥线以下 PIPE16 单元
NSEL,S,LOC,Z,7.9        !选中 Z=7.9 处的节点
NSEL,A,LOC,Z,-14.6       !继续选中 Z=-14.6 处的节点
NSEL,A,LOC,Z,-38.6       !继续选中 Z=-38.6 处的节点
NSEL,A,LOC,Z,-63.6       !继续选中 Z=-63.6 处的节点
!NSEL,A,LOC,Z,-90.6      !继续选中 Z=-90.6 处的节点（选择性约束）
D,ALL,UX                !约束所选节点 X 方向位移自由度
D,ALL,UY                !约束所选节点 Y 方向位移自由度
ALLSEL                  !选中所有对象
SAVE                    !保存模型
FINISH                  !退出前处理
!*****求解计算过程
/SOLU                   !进入求解计算模块
ANTYPE,0                !设置求解类型为结构静力分析
NSEL,S,LOC,Z,-120.63     !选中 Z=-120.63 处的节点
D,ALL,ALL                !约束所选节点 X 方向位移自由度
ALLSEL                  !选中所有对象
NSEL,S,LOC,Z,38          !选中 Z=38 处的节点
D,ALL,UX                 !约束所选节点 X 方向位移自由度
D,ALL,UY                 !约束所选节点 Y 方向位移自由度
ALLSEL                  !选中所有对象
NSEL,S,LOC,Z,38          !选中 Z=38 处的节点
F,ALL,FZ,F_top           !施加井口压载

```



```

ALLSEL                !选中所有对象
NSEL,S,LOC,Z,1,38     !选中 Z=1~38 范围内的所有节点
F,ALL,FX,F_wind/38    !施加风载
ALLSEL                !选中所有对象
ACEL,0,0,9.8          !施加重力载荷
SOLVE                 !执行求解
SAVE                  !保存求解结果
FINISH                !退出求解计算模块
!*****静力分析后处理过程
/POST1                !进入通用后处理器
SET,LAST              !读入计算结果文件
/ESHAPE,1.0           !打开模型形状开关
PLDISP,0              !绘制结构整体变形图
PLNSOL,U,SUM,0,1.0     !绘制结构位移等值图
PLNSOL,S,EQV,0,1.0     !绘制结构等效应力分布云图
PLNSOL,EPTO,EQV,0,1.0  !绘制结构等效应变分布云图
ETABLE, MOMENT_I,SMISC,5 !定义 I 端节点弯矩
ETABLE, MOMENT_J,SMISC,11 !定义 J 端节点弯矩
PLLS,MOMENT_I,MOMENT_J,1,0 !绘制弯矩图
ETABLE, STRESS_I,NMISC,15 !定义 I 端节点应力
ETABLE, STRESS_J,NMISC,55 !定义 J 端节点应力
PLLS, STRESS_I, STRESS_J,1,0 !绘制应力分布图
FINISH                !退出后处理器
!*****模态分析过程
/PREP7                !重新进入前处理
ET,3,MASS21           !定义 3 号单元类型为 MASS21
R,3,,3E4              !定义 3 号实常数（针对 MASS21 单元）
TYPE,3                !选择 3 号单元类型
REAL,3                !选择 3 号实常数
E,1                    !通过节点 1 建立 MASS21 实体单元
NSEL,S,LOC,Z,-120.6,-90 !选中底端与水下第四层导向孔之间的所有节点
NSEL,A,LOC,Z,-90,-64    !继续选中水下第四层导向孔与第三层之间的所有节点
NSEL,A,LOC,Z,-63,-39    !继续选中水下第三层导向孔与第二层之间的所有节点
NSEL,A,LOC,Z,-38,-15    !继续选中水下第二层导向孔与第一层之间的所有节点
NSEL,A,LOC,Z,-14,7      !继续选中水下第一层导向孔与水上导向孔之间的所有节点
NSEL,A,LOC,Z,8,37       !继续选中水上导向孔与顶部的所有节点

```

```

D,ALL,UY          !约束所选节点 Y 方向的位移自由度
D,ALL,UZ          !约束所选节点 Z 方向的位移自由度
D,ALL,ROTX        !约束所选节点 X 方向的转动自由度
D,ALL,ROTZ        !约束所选节点 Z 方向的转动自由度
ALLSEL            !选中所有对象
NSEL,Z,LOC,Z,-120.63 !选中底端节点
D,ALL,ALL          !约束其全部自由度
ALLSEL            !选中所有对象
ACEL,0,0,9.8       !施加重力载荷
/SOLU              !进入求解模块
ANTYPE,MODAL       !设置分析类型为模态分析
MODEOPT,LANB,6     !模态分析选项
MXPAND,6           !设置扩展 4 阶模态
SOLVE              !开始模态求解
SAVE               !保存求解结果
FINISH             !退出求解模块
!*****模态分析后处理
/POST1             !进入通用后处理器
ESHAPE,1.0         !打开结构形状开关
SET,LIST           !列表显示结构自振频率计算结果
SET,FIRST          !读入一阶模态
PLDISP,0           !显示结构一阶模态
SET,NEXT           !读入二阶模态
PLDISP,0           !显示结构二阶模态
SET,NEXT           !读入三阶模态
PLDISP,0           !显示结构三阶模态
SET,NEXT           !读入四阶模态
PLDISP,0           !显示结构四阶模态
FINISH             !退出后处理模块

```

## 第9章 导管架平台整体结构分析

### ◇ 本章导读

本章以海洋石油导管架平台为对象,采用 GUI 菜单操作和命令流相结合的方式,详细介绍了 ANSYS 建立海洋平台有限元模型的全过程,针对海洋平台的各种环境载荷,采用相应的 ANSYS 分析类型分别进行计算,目的是希望通过具体实例,加深读者对 ANSYS 结构整体分析的理解,以及对复杂海洋结构物分析的方法。

本章包括如下的一些问题:

- 平台结构模型的建立
- 海洋平台整体结构的静力分析
- 海洋平台的模态分析
- 波浪载荷作用下海洋平台的瞬态动力分析

海洋平台是高出海平面的一种海洋工程结构。前已述及,海洋平台按结构类型分为固定式平台和移动式平台。固定式平台又可以分为导管架型、塔形和重力型等各种结构形式。图所示是导管架平台的一些工程实例。移动式平台则包括自升式、半潜式、浮船式和张力腿式等结构形式。由于海洋平台工作环境是在近海海面上,受到风浪流等载荷的作用,因此对其安全性和可靠性的分析是确保其在服役年限内正常使用的重要环节。本章的目标就是通过一个 ANSYS 导管架平台的分析实例,使得读者能够了解和学习固定式导管架海洋平台及类似的结构建模和有限元分析方法。导管架平台如图 9-1 所示。



图 9-1 导管架平台

## 9.1 平台整体模型建立

本节结合工程实例以导管架平台为对象,按照结构特点分析、单元类型选择、材料参数设定、建立几何模型、划分有限元网格的顺序,详细介绍 ANSYS 建立海洋导管架平台有限元模型的过程。建模过程中,采用 APDL 命令流和 GUI 菜单相结合的方式,目的是使读者具体体会这两种建模方式各自的特点。

### 9.1.1 工程实例基本数据

#### 9.1.1.1 环境条件

水深: 45m

风速: 43.6m/s

有效波高: 14.8m, 有效波周期: 10.8s,

海面流速: 2.35m/s, 中部流速: 1.96m/s, 底部流速: 1.60m/s。

海冰: 厚度: 32cm, 抗压强度 1990Kpa。

#### 9.1.1.2 平台结构几何参数

平台结构几何参数见表 9.1。

表 9.1 平台几何参数 单位: m

序号	名称	直径		壁厚	
1	泥面上导管架主导管	1.2		0.050	
2	泥面下导管架主导管	1.2		0.050	
3	甲板部分主导管	0.78		0.0381	
3	导管架层间横支撑	0.78		0.0381	
4	导管架斜撑杆	0.508		0.0254	
5	甲板部分梁结构	截面面积	截面长	截面宽	
		0.16	0.4	0.4	
6	甲板	厚度		长*宽	
		0.025		30*20	

#### 9.1.2 导管架平台结构特点

本节主要是建立如图 9-2 所示的四腿导管架平台。




图 9-2 导管架平台模型

### (1) 总体特点

导管架平台属于固定式平台，由上层平台结构和下部导管架结构组成，导管架底端通过桩基础固定。上层平台结构支撑框架和甲板，主要提供生产和生活的场地，其外形为矩形，尺寸为： $30m \times 20m$ ，水面至顶层甲板高度为 15m。

下部导管架由一系列钢管焊接而成，主体是 4 根主导管，其间用细管件作为撑杆，组成空间塔架结构，桩基础通过主导管插入海底泥土。

### (2) 单元类型选择和参数设定

整个导管架平台模型采用四种单元类型：PIPE20、PIPE59、BEAM4 及 SHELL43。下部导管架和上部甲板框架的主要竖向支撑构件采用 PIPE59 单元，泥线以下采用 PIPE20 单元，甲板平面的框架梁采用 BEAM4 单元，水平甲板采用 SHELL43 单元。

1) PIPE20 单元截面由外径和壁厚确定；

2) PIPE59 单元截面由外径、壁厚、法向曳力系数、法向惯性系数、内部液体密度、外部附着物密度、外部附着物厚度确定；

3) BEAM4 单元截面形式选用矩形，单元参数包括截面的高度、宽度、面积和截面惯性矩；

4) SHELL43 单元参数包括四节点处的厚度。

整个模型采用同一种钢材，弹性模量为  $2.0E11Pa$ ，泊松比为 0.3，密度为  $7850kg/m^3$  材质为 D36，屈服强度为 360Mpa。

### (3) 建模步骤

结构的建模步骤主要分为几何模型的建立和划分网格建立有限元模型。

1) 建立几何模型, 按照平台各节点的具体坐标建立关键点, 连接关键点生成线, 从而建立平台结构的基本框架, 创建甲板平面, 完成几何模型的建立。

2) 建立有限元模型, 对几何模型进行网格划分, 形成计算分析的有限元模型。

## 9.1.2 平台几何模型的建立

本节主要以命令流方式进行介绍, 同时也给出相应的 GUI 操作的菜单路径, 建模过程中, 所有数据输入统一采用国际单位制单位。下面对建立几何模型的各步骤及其注意事项进行具体介绍, 同时给出一些关键步骤和程序执行结果。

### (1) 设置工作环境

进入 ANSYS/Multiphysics 的程序界面后, 通过常用菜单路径【Utility Menu】>【File】>【Change Jobname】, 指定分析的工作名称为“Platform”, 将“New log and error files?”选项设置为【yes】, 点击【OK】按钮, 通过菜单项【Utility Menu】>【File】>【Change Title】, 指定图形显示区域的标题为“Analysis of Platform”。在命令流窗口输入“/UNIT,SI”, 定义国际单位制。

### (2) 定义单元类型

单击菜单路径【Main Menu】>【Preprocessor】>【Element Type】>【Add/Edit/Delete】定义 4 种单元类型, 选取 PIPE20 为 1 号单元, PIPE59 为 2 号单元, BEAM4 为 3 号单元, SHELL43 为 4 号单元。操作命令如下:

ET,1,PIPE20	!定义 1 号单元 PIPE20
ET,2,PIPE59	!定义 2 号单元 PIPE59
ET,3,BEAM4	!定义 3 号单元 BEAM4
ET,4,SHELL43	!定义 4 号单元 SHELL43

### (3) 定义单元实常数

单击菜单路径【Main Menu】>【Preprocessor】>【Real Constants】>【Add/Edit/Delete】, 定义 6 组实常数, 其中实常数 1 针对 PIPE20 单元, 实常数 2~4 针对 PIPE59 单元, 实常数 5 针对 BEAM4 单元, 实常数 6 针对 SHELL63 单元。命令流如下:

R,1,1.2,0.050	!定义 1 号实常数 (PIPE20)
R,2,1.2,0.050, 0.7,2.0,1030,0	!定义 2 号实常数 (PIPE59)
R,3,0.78,0.0381, 0.7,2.0,1030,0	!定义 3 号实常数 (PIPE59)
R,4,0.508,0.0254, 0.7,2.0,1030,0	!定义 4 号实常数 (PIPE59)
R,5,0.06,0.0002,0.00045,0.3,0.2	!定义 5 号实常数 (BEAM4)

R,6,0.02,0.02,0.02,0.02

!定义 6 号实常数 (SHELL43)

#### (4) 定义材料参数

选择菜单项【Main Menu】>【Preprocessor】>【Material Props】>【Material Models】，在出现的“Define Material Model Behavior”对话框的右侧，依次选择菜单路径项【Structural】>【Nonlinear】>【Inelastic】>【Rate Independent】>【Isotropic Mardening Plasticity】>【Mises Plasticity】>【Bilinear】，弹出“Note”对话框，点击【确定】按钮，设置材料的线性阶段的弹性模量及泊松比，在出现的对话框【EX】项后的文本框中输入“2.1E11”，在【PRXY】项后的文本框中输入“0.3”。然后单击【OK】按钮，弹出如图 6-8 所示的材料屈服强度设置窗口，在【Yield Stss】项后的文本框中输入“360E6”，在【Tang Mod】项后的文本框中输入“0”，表示采用的材料为理想弹塑性模型。再次单击【Structural】>【Density】，弹出密度设置窗口，在【DENS】选项后的文本框中输入“7850”，单击【OK】。继续单击【Fluids】>【Water Table】设置环境载荷参数，这部分的定义方法与第 4 章中相同，再此不再赘述，具体设置见本章附录。

#### (5) 几何模型建立

建立海洋平台的几何模型主要分为三个步骤，即创建关键点、连线构成结构框架和生成甲板平面。下面将按照上述三个步骤介绍海洋平台几何模型的创建过程。

##### 1) 创建关键点

通过菜单路径【Main Menu】>【Preprocessor】>【Modeling】>【Create】>【Keypoint】>【In Active CS】创建关键点。操作命令如下：

K,NPT,X,Y,Z

!通过定义三个坐标值生成关键点

完整的关键点创建过程见本章附录部分命令流，建立关键点之后的模型如图 9-3 所示。




图 9-3 平台关键点



## 2) 通过关键点创建线构成结构框架

连接关键点生成平台结构的杆系框架，通过菜单路径【Main Menu】>【Preprocessor】>【Modeling】>【Create】>【Lines】>【Straight Line】由关键点创建线。操作命令流如下：

```
L,P1,P2          !通过关键点创建线
```

完整的线创建见命令流附件。通过关键点创立线框架之后的模型如图 9-4 所示。

## 3) 生成甲板平面

每层甲板平面由 9 个矩形平面组成，矩形位置和尺寸由甲板关键点坐标确定。通过菜单路径【Main Menu】>【Preprocessor】>【Modeling】>【Create】>【Area】>【Arbitrary】>【Through Kps】。操作命令流如下：

```
A,P1,P2,P3,P4    !由关键点生成面
```

完整的平面创建过程详见命令流附件。生成甲板平面之后的模型如图 9-5 所示。

到此为止，海洋平台的几何模型创建完毕，在 ANSYS Toolbar 上单击【SAVE\_DB】按钮，保存模型到数据文件。




图 9-4 模型框架




图 9-5 生成甲板面

## 9.1.3 平台有限元模型划分

在第 2 步的单元类型定义中，定义了 PIPE20、PIPE59、BEAM4、SHELL43 四种单元类型，分别用来划分导管架、甲板平面梁框架和甲板，下面按照这四种单元的顺序对结构模型进行划分网格生成有限元模型。

### 1) 导管架网格划分

PIPE20 和 PIPE59 都是用来划分导管架结构，PIPE20 选用 1 号材料属性，单元截面由外径和壁厚确定；PIPE59 采用 2 号材料属性，由外径、壁厚及一系列流体参数确定单元属性。不同规格的杆件需要结合相应的实常数，所有被划分线段的网格划分数 NDIV 为 1。通过菜单路径【Main Menu】>【Preprocessor】>【Meshing】>【MeshTool】。操作的命令流如下：

```
LSEL,S,,,VMIN,VMAX    !通过线号选取线
```



LATT,MAT,REAL,TYPE	!为已选取的线赋予材料、实常数、单元类型编号
LESIZE,ALL,, NDIV	!对已选取的线划分份数
LMESH,ALL	!网格划分

## 2) 甲板平面梁网格划分

BEAM4 单元用来划分甲板梁。所有 BEAM4 单元采用 1 号材料属性, 5 号实常数, 线段网格划分段数为 3。通过菜单路径【Main Menu】>【Preprocessor】>【Meshing】>【MeshTool】。命令流同导管架网格划分。

## 3) 甲板平面网格划分

SHELL63 单元用来划分甲板平面。所有 SHELL63 单元采用 1 号材料属性, 6 号实常数。通过菜单路径【Main Menu】>【Preprocessor】>【Meshing】>【MeshTool】。操作命令流如下:

ASEL,S,,,ALL	!选取面
AATT,1,6,3	!为已选取的面赋予材料、实常数、单元类型编号
AMEH,ALL	!面单元划分

至此, 海洋平台结构的有限元模型建立完毕。

为了便于观察, 选取菜单路径【Utility Menu】>【PlotCtrls】>【Style】>【Size and Shape】, 弹出“Size and Shape”对话框, 打开“Display of element shapes on real constant descriptions”选项, 模型将以实际形状显示。如图 9-6 所示。图中不同颜色代表不同的实常数。在 ANSYS Toolbar 上单击【SAVE\_DB】, 保存模型到数据文件。




图 9-6 导管架平台整体有限元模型

## 9.2 波流耦合作用下导管架平台整体结构静力分析

本节将在上一节建立的平台有限元模型基础上对平台进行静动力计算分析，所进行的是非线性求解，并将计算所得相应数据进行校核。模型的静力分析是利用前述相关章节所给出的荷载理论，考虑平台在风浪流作用下的结构静力分析。

### 9.2.1 波流耦合力极值

在对导管架平台进行整体静力分析之前，首先要对该结构进行波流耦合力极值进行搜索，以求得相对应的波浪相位角。对于波浪相位角的搜索，一般是针对某一固定波高和周期的波浪而言的。当已知的波浪作用于结构物时，其作用力的大小与相位角紧密相关。按照前述相关章节所讲述的方法对平台整体结构进行相位角搜索。

(1) **模型导入：**读入上节所建立的导管架有限元模型到 ANSYS 程序；

(1) **设置环境参数：**假设波流均沿 X 正向作用于平台结构，将波浪相位角置空；

(2) **设置边界条件：**为了方便结构反力的提取，将结构最底端四节点进行自由度耦合并约束其中一个节点。见图 9-7 所示。




图 9-7 施加重力载荷及约束之后的模型

(3) **进行搜索求解：**利用第四章有关程序进行求解。

(4) **提取分析结果：**根据分析结果数据做出波流耦合力—相位角关系曲线，如图 9-8 所示。结合结果文件数据，可得出当波流耦合力最大值为 4913KN，相对应的相位角为 46°。此相位角即

作为静力分析时所输入的值。




图 9-8 波流耦合力—相位角关系曲线

## 9.2.2 结构整体静力分析

### (1) 设置分析类型

选择菜单路径【Main Menu】>【Solution】>【Analysis Type】>【New Analysis】，选择“Static”，点击【OK】按钮退出。

### (2) 定义位移边界条件

选择菜单路径【Main Menu】>【Solution】>【Define Loads】>【Apply】>【Structural】>【Displacement】>【On Nodes】，弹出节点拾取对话框，在图形显示区域用鼠标将导管架底端四节点全部选上，单击【OK】按钮，弹出位移属性设置对话框，选中【All DOFS】，单击【OK】按钮退出。

### (3) 施加载荷

1) 施加波流载荷：按波流同方向作用于结构物，波浪相位角为  $46^\circ$ 。

2) 定义重力场：

单击菜单【Main Menu】>【Solution】>【Define Loads】>【Apply】>【Structural】>【Inertia】>【Gravity】>【Global】，弹出“Apply (Gravitational) Acceleration”属性值设置对话框，在“ACELX, ACELY, ACELZ”中依次输入“0, 0, 9.8”。

### (4) 静力分析求解

通过菜单项【Main Menu】>【Solution】>【Solve】>【Current LS】，对问题进行求解。在求解过程中会看到如图所示的求解收敛曲线。在求解结束后，弹出 Solution is done! 信息提示框，关闭。单击【Main Menu】>【Finish】菜单项，退出求解计算模块。

### 9.2.3 静力结果分析

#### 1) 读入结果文件

选择菜单路径【Main Menu】>【General Postproc】>【Read Results】，单击【Last Set】，读入最后一个子步的结果文件。

#### 2) 绘制结构整体变形图

选择菜单路径【Main Menu】>【General Postproc】>【Plot Results】>【Deformed Shape】，在弹出的对话框中选择【Def shape only】，显示结构整体变形图，如图 9-9 所示。




图 9-9 结构整体变形图

#### 3) 绘制结构节点位移等值线云图

选择菜单路径【Main Menu】>【General Postproc】>【Plot Results】>【Contour Plot】>【Nodal Solu】，弹出“Contour Nodal Solution Data”对话框，依次单击【Nodal Solution】>【DOF Solution】>【Displacement vector sum】，单击【OK】按钮，显示如图 9-10 所示的结构节点位移等值线云图。从图中可以看出，结构发生最大位移的位置位于平台顶部，数值为 0.0398m。




图 9-10 节点位移等值线云图

## 4) 绘制单元应力等值线云图

选择菜单路径【Main Menu】>【General Postproc】>【Plot Results】>【Contour Plot】>【Element Solu】，弹出“Contour Element Solution Data”对话框，依次单击【Element Solution】>【Stress】>【von Mises stress】，单击【OK】按钮，如图 9-11 所示显示结构的米塞斯等效应力云图。由图可知，结构的最大应力为 117MPa，发生在桩腿底部位置。




图 9-11 单元等效应力云图

## 9.3 导管架平台整体结构模态分析

模态分析是用来确定结构振动特性的一种技术，主要是求解结构的自然频率、振型以及振型参与系数等，它是所有动力学分析类型的最基础的内容，其分析结果可以作为瞬态动力分析、谐响应分析和谱分析等其他动力分析的基础。因此本节将对所建立的导管架平台进行模态分析。

### 9.3.1 结构模态计算

#### (1) 设定模态分析类型

重新进入 ANSYS 求解器，选择菜单路径【Main Menu】>【Solution】>【Analysis Type】>【New Analysis】，设置分析类型为【Modal】。

#### (2) 设置分析选项

选择菜单路径【Main Menu】>【Solution】>【Analysis Type】>【Analysis Option】，弹出“Modal Analysis”对话框，模态提取方法采用【Block Lanczos】，提取模态数设为“6”。

#### (3) 扩展模态设置

选择菜单路径【Main Menu】>【Solution】>【Load Step Opts】>【Expansion Pass】>【Single Expand】>【Expand Modes】，弹出“Expand Nodes”对话框，模态扩展数设为“6”。

#### (4) 定义位移边界条件

选择菜单路径【Main Menu】>【Solution】>【Define Loads】>【Apply】>【Structural】>【Displacement】>【On Nodes】，弹出节点拾取对话框，在图形显示区域用鼠标将导管架底端四节点全部选上，单击【OK】按钮，弹出位移属性设置对话框，选中【All DOFS】，单击【OK】按钮退出。


#### (5) 模态分析求解

选择菜单路径【Main Menu】>【Solution】>【Solve】>【Current LS】，弹出“Solve Current Load Step”对话框，单击【OK】，开始计算模态解。求解完毕后，在“Note”窗口显示“Solution is done! ”，单击【Close】关闭窗口。

### 9.3.2 观察模态分析结果

#### 1) 观察模型固有频率

进入通用后处理器 POST1，选择菜单路径【Main Menu】>【General Postproc】>【Results Summary】，观察模型的固有频率，结果如图 9-12 所示。



\*\*\*\*\* INDEX OF DATA SETS ON RESULTS FILE \*\*\*\*\*

SET	TIME/FREQ	LOAD STEP	SUBSTEP	CUMULATIVE
1	1.2514	1	1	1
2	1.2560	1	2	2
3	1.3508	1	3	3
4	3.0395	1	4	4
5	3.2496	1	5	5
6	3.4074	1	6	6

图 9-12 平台结构前六阶固有频率

## 2) 读入结果文件

选择菜单路径【Main Menu】>【General Postproc】>【Read Results】>【First Set】，读入第 1 载荷子步的计算结果。

## 3) 观察模态振型

选择菜单路径【Main Menu】>【General Postproc】>【Plot Results】>【Deformed Shape】，显示一阶模态振型，如图 9-13 所示。

为了便于观察，建议选择菜单路径【Utility Menu】>【PlotCtrls】>【Style】>【Size and Shape】，在“Size and Shape”对话框打开“Display of element shapes based on real constant descriptions”选项，以便在图形窗口以单元实际截面形状显示模型的模态振型。

重复上述操作可以得出其他五阶模态振型，如图 9-14 至 9-18 所示。




图 9-13 一阶模态




图 9-14 二阶模态




图 9-15 三阶模态




图 9-16 四阶模态




图 9-17 五阶模态




图 9-18 六阶模态

模态分析部分命令流如下:

/SOLU	!进入求解模块
ANTYPE,MODAL	!设置分析类型为模态分析
MODOPT,LANB,6	!模态分析选项
MXPAND,6	!设置扩展 6 阶模态
NSEL,S,LOC,Z,-50	!选中 Z=-50 处的节点
D,ALL,ALL	!约束所选节点全部自由度
ALLSEL	!选中所有对象
ACEL,0,0,9.8	
SOLVE	!开始模态求解
SAVE	!保存求解结果
FINISH	!退出求解模块
!*****模态分析后处理	
/POST1	!进入通用后处理器
ESHAPE,1.0	!打开结构形状开关



SET,LIST	!列表显示结构自振频率计算结果
SET,FIRST	!读入一阶模态
PLDISP,0	!显示结构一阶模态
SET,NEXT	!读入二阶模态
PLDISP,0	!显示结构二阶模态
SET,NEXT	!读入三阶模态
PLDISP,0	!显示结构三阶模态
SET,NEXT	!读入四阶模态
PLDISP,0	!显示结构四阶模态
SET,NEXT	!读入五阶模态
PLDISP,0	!显示结构五阶模态
SET,NEXT	!读入六阶模态
PLDISP,0	!显示结构六阶模态
FINISH	!退出后处理模块

## 9.4 波浪作用下平台结构瞬态动力分析

瞬态动力学分析又称时间历程分析，用于计算结构在随时间任意变化的载荷作用下的动力学响应，目的是得到结构在稳态载荷、瞬态载荷和简谐载荷随意组合作用下随时间变化的位移、应变、应力和力。

波浪力是一种随时间变化的载荷，对导管架平台的动力响应问题不可忽视。本节将进行波浪载荷作用下导管架平台的整体瞬态动力分析。

### 9.4.1 瞬态动力分析

进行瞬态动力分析所采用的结构有限元模型与模态分析时完全相同。由模态分析所得到的阻尼系数进行瞬态动力分析，时程分析取作用时间为 100s，时间间隔 0.2s。根据模态分析所得前两阶固有频率可计算得到瑞利阻尼系数为： $\alpha=0.1575$ ， $\beta=2.5389\text{E-}03$ 。

#### (1) 波流参数修改

由于是分析波载载荷所引起的水动力效应，因此将波浪的相位角设置为 0 度，选择【Main Menu】>【Preprocessor】>【Material Props】>【Material Models】，在弹出的“Define Material Model Behavior”对话框中，选择【Water Table】选项，单击打开，将原来静力分析所输入的波浪相位角 46 度改为 0 度，单击【OK】按钮退出。命令流如下：

TB,81,0	!修改 WATERTABLE 里的波浪相位角为 0°
---------	----------------------------

## (2) 求解选项设置

### 1) 设置求解类型

选择菜单路径【Main Menu】>【Solution】>【Analysis Type】>【New Analysis】，弹出“Type of analysis”对话框，选中分析类型为【Transient】，分析方法选择【Full】，单击【OK】按钮退出。

### 2) 求解控制项设置

选择菜单路径【Main Menu】>【Solution】>【Analysis Type】>【Sol'n Controls】，弹出“Solution Controls”对话框，在【Basic】选项下的【Analysis Options】选中“Large Displacement Transient”，然后选中【Transient】，在【Damping Coefficients】选项下的【ALPHA】后输入“0.1575”，在【BETA】后输入“2.5389E-03”，其他选项默认，单击【OK】按钮退出，如图 7-11 所示。命令流如下：

ANTYPE,TRANS	!设置求解类型为瞬态动力分析
ALPHAD, 0.1575	!设置阻尼系数 $\alpha$
BETAD, 2.5389E-03	!设置阻尼系数 $\beta$

## (3) 定义位移边界条件

选择菜单路径【Main Menu】>【Solution】>【Define Loads】>【Apply】>【Structural】>【Displacement】>【On Nodes】，弹出节点拾取对话框，在图形显示区域用鼠标将导管架底端四节点全部选上，单击【OK】按钮，弹出位移属性设置对话框，选中【All DOFS】，单击【OK】按钮退出。

## (4) 施加重力载荷

单击菜单【Main Menu】>【Solution】>【Define Loads】>【Apply】>【Structural】>【Inertia】>【Gravity】>【Global】，弹出“Apply (Gravitational) Acceleration”属性值设置对话框，在“ACELX, ACELY, ACELZ”中依次输入“0, 0, 9.8”。

## (5) 求解

设置时程分析结束时间为 100s，时间间隔为 0.2s。本部分利用命令流形式输入，如下：

*DO,I,10E-12,100,0.2	!设定结束时间及子步时间开始循环计算
TIME,I	!循环计算结束时间
OUTPES,ALL,ALL	!输出结果设置
SOLVE,SAVE	!执行求解并保存
*ENDDO	!结束循环求解计算

## 9.4.2 动力分析结果处理

### 1) 进入时间历程后处理器 POST26

选择菜单路径【Main Menu】>【TimeHist Postpro】，进入时间历程后处理器

### 2) 定义时间历程后处理变量 2

选择【Main Menu】>【TimeHist Postpro】>【Define Variables】，在弹出的“Define Time-History Variables”对话框中，单击【Add】按钮，出现“Add Time-History Variables”对话框，选中【Nodal DOF Result】，单击【OK】按钮，出现的“Define Nodal Data”对象拾取框中，选中【List of Items】，在文本框中输入“1173”，即平台顶端甲板中心，然后单击【OK】按钮。出现“Define Nodal Data”对话框，在【Name】一栏中输入“UX”，在【Item, Comp】选项后的列表中选择【DOF solution】>【Translation UX】，其他选项默认，单击【OK】退出。

### 3) 绘制时程—位移曲线

选择菜单路径【Main Menu】>【TimeHist Postproc】>【Settings】>【Graph】，定义横向（X）坐标为时间，在“Graph Settings”对话框中的【XVAR】(X-axis variables)选项后选中【Single variable】，在【Single variable No.】文本框中输入“1”，单击【OK】按钮退出该对话框。

选择菜单路径【Utility Menu】>【PlotCtrls】>【Style】>【Graphs】>【Modify Axes】，在弹出的对话框中输入图形显示的横、纵坐标的标题，分别为“TIME”和“UX”，单击【OK】按钮退出。

选择菜单路径【Main Menu】>【TimeHist Postproc】>【Graph Variables】，弹出对话框中，“NVAR1”文本框中输入“2”，单击【OK】按钮，则可绘制时间—位移曲线，如图 9-19 所示。




图 9-19 桩腿顶端时程—位移曲线

### 4) 绘制结构主导管腿处反力—时程曲线

选择【Main Menu】>【TimeHist Postpro】>【Define Variables】，在弹出的“Define Time-History

Variables”对话框中，单击【Add】按钮，出现“Add Time-History Variables”对话框，选中【Reaction force】，单击【OK】按钮，弹出对象拾取框中，选中【List of Items】，在文本框中输入“125”，即主导管处节点编号，然后单击【OK】按钮。

选择菜单路径【Utility Menu】>【PlotCtrls】>【Style】>【Graphs】>【Modify Axes】，在弹出的对话框中输入图形显示的横、纵坐标的标题，分别为“TIME”和“Reforce”，单击【OK】按钮退出。

选择菜单路径【Main Menu】>【TimeHist Postproc】>【Graph Variables】，弹出对话框中，“NVAR1”文本框中输入“3”，单击【OK】按钮，则可绘制时间—反力曲线，如图 9-20 所示。




图 9-20 主导管之一时程—反力曲线

#### 5) 绘制结构弱点处的应力—时程曲线

选择【Main Menu】>【TimeHist Postpro】>【Define Variables】，在弹出的“Define Time-History Variables”对话框中，单击【Add】按钮，出现“Add Time-History Variables”对话框，选中【...by set no】，单击【OK】按钮，弹出对象拾取框中，选中【List of Items】，在文本框中输入“125”，即连接点处节点编号，然后单击【OK】按钮。出现“Define Element Results by Seq No.”对话框，在【Name】一栏中输入“STRESS”，在【Item】选项后的列表中选择【SMISC】，在【Comp Sequence number】一栏中输入“15”，其他选项默认，单击【OK】退出。

选择菜单路径【Utility Menu】>【PlotCtrls】>【Style】>【Graphs】>【Modify Axes】，在弹出的对话框中输入图形显示的横、纵坐标的标题，分别为“TIME”和“STRESS”，单击【OK】按钮退出。

选择菜单路径【Main Menu】>【TimeHist Postproc】>【Graph Variables】，弹出对话框中，

“NVAR1”文本框中输入“4”，单击【OK】按钮，则可绘制时间—应力曲线，如图 9-21 所示。




图 9-21 结构应力—时程曲线

## 9.6 本章小结

本章主要介绍了海洋导管架平台的基本结构分析知识，主要包括平台的模型建立，结构整体静力分析，以及模态分析方法和平台在波浪载荷作用下的动力分析方法。希望通过本章的介绍，读者可以掌握有关海洋平台的一些结构分析方法。

### 附录 6：本章实例 APDL 建模及分析命令流

```
!*****
!*****
!*****海洋导管架平台整体结构分析*****
!*****
!*****
/FILNAME, Platform,1    !设置工作文件名
/TITLE,Analysis of Platform    !设置图形显示区域名
!*****定义环境载荷参数
Dwater=45                !水深
H_wave=14.8              !波高
T_wave=10.8              !波周期周期
V_surface=2.35           !水面处海流速度
V_middle=1.96            !中部海流速度
V_bottom=1.60            !底部海流速度
/PREP7                   !进入前处理
ET,1,PIPE20              !定义 1 号单元 PIPE20
```

```

ET,2,PIPE59      !定义 2 号单元 PIPE59
ET,3,BEAM4       !定义 3 号单元 BEAM4
ET,4,SHELL43     !定义 4 号单元 SHELL43
R,1,1,2,0.050    !定义 1 号实常数 (PIPE20)
R,2,1,2,0.050, 0.7,2.0,1030,0    !定义 2 号实常数 (PIPE59)
R,3,0.78,0.0381, 0.7,2.0,1030,0    !定义 3 号实常数 (PIPE59)
R,4,0.508,0.0254, 0.7,2.0,1030,0    !定义 4 号实常数 (PIPE59)
R,5,0.16,0.0021,0.0021,0.4,0.4    !定义 5 号实常数 (BEAM4)
R,6,0.025,0.025,0.025,0.025    !定义 6 号实常数 (SHELL43)
MP,EX,1,2,1E11   !设置弹性模量
MP,PRXY,1,0.3    !设置泊松比
MP,DENS,1,7850   !设置密度
TB,BKIN,1,1      !塑性设置
TBTEMP,0
TBDATA,1,360E6,0    !屈服强度设置
!*****波流参数 Water Table
TBDE,WATE,1
TB,WATE,1,,,0
TBMODIF,1,1,2    !选择 Stokes 五阶波理论
TBMODIF,1,2,0    !选择波流耦合作用方式
TBMODIF,1,3, Dwater !水深
TBMODIF,1,4,1030 !海水密度
TBMODIF,1,5,0    !波浪作用方向
TBMODIF,2,1,- Dwater !海水底部坐标值
TBMODIF,2,2,V_bottom !底部流速
TBMODIF,2,3,0    !底部海流方向
TBMODIF,2,4,- Dwater/2 !海水中部坐标值
TBMODIF,2,5,V_middle !中部流速
TBMODIF,2,6,0    !中部海流方向
TBMODIF,3,1,0    !海水表面坐标值
TBMODIF,3,2,V_surface !表面流速
TBMODIF,3,3,0    !表面海流方向
TBMODIF,14,1,H_wave !波高
TBMODIF,14,2,T_wave !周期
TBMODIF,14,3,0    !波浪相位角 (不同分析需设置不同值)
!*****几何模型建立
!*****主导管关键点
K,1,6,6,0

```

```
K,6,10,10,-50
KFILL,1,6,4,2,1,1,
K,7,6,-6,0
K,12,10,-10,-50
KFILL,7,12,4,8,1,1,
K,13,-6,6,0
K,18,-10,10,-50
KFILL,13,18,4,14,1,1,
K,19,-6,-6,0
K,24,-10,-10,-50
KFILL,19,24,4,20,1,1,
!*****泥线以下主导管
KFILL,5,6,1,25,,,
KFILL,11,12,1,26,,,
KFILL,17,18,1,27,,,
KFILL,23,24,1,28,,,
!*****导管架中心点
K,29,0,0,0
K,34,0,0,-50
KFILL,29,34,4,30,1,1,
KFILL,33,34,1,35,,,
!*****上部甲板处主关键点
!*****第一层甲板
K,36,6,6,8
K,37,6,-6,8
K,38,-6,-6,8
K,39,-6,6,8
K,40,15,10,8
K,41,15,-10,8
K,42,-15,-10,8
K,43,-15,10,8
K,44,15,6,8
K,45,15,-6,8
K,46,6,-10,8
K,47,-6,-10,8
K,48,-15,-6,8
K,49,-15,6,8
K,50,-6,10,8
K,51,6,10,8
K,52,0,0,8

K,53,15,0,8
K,54,10,-10,8
K,55,0,-10,8
```

K,56,-10,-10,8  
K,57,-15,0,8  
K,58,-10,10,8  
K,59,0,10,8  
K,60,10,10,8  
K,61,10,6,8  
K,62,10,0,8  
K,63,10,-6,8  
K,64,-10,-6,8  
K,65,-10,0,8  
K,66,-10,6,8  
K,67,0,6,8  
K,68,6,0,8  
K,69,0,-6,8  
K,70,-6,0,8  
!\*\*\*\*\*第二层甲板  
K,71,6,6,15  
K,72,6,-6,15  
K,73,-6,-6,15  
K,74,-6,6,15  
K,75,15,10,15  
K,76,15,-10,15  
K,77,-15,-10,15  
K,78,-15,10,15  
K,79,15,6,15  
K,80,15,-6,15  
K,81,6,-10,15  
K,82,-6,-10,15  
K,83,-15,-6,15  
K,84,-15,6,15  
K,85,-6,10,15  
K,86,6,10,15  
K,87,0,0,15  
K,88,15,0,15  
K,89,10,-10,15  
K,90,0,-10,15  
K,91,-10,-10,15  
K,92,-15,0,15  
K,93,-10,10,15  
K,94,0,10,15  
K,95,10,10,15  
K,96,10,6,15  
K,97,10,0,15  
K,98,10,-6,15



```
K,99,-10,-6,15
K,100,-10,0,15
K,101,-10,6,15
K,102,0,6,15
K,103,6,0,15
K,104,0,-6,15
K,105,-6,0,15
!*****建立线框架
!*****主导管
LSTR,1,2      !1 号线
LSTR,2,3      !2 号线
LSTR,3,4      !3 号线
LSTR,4,5      !4 号线
LSTR,5,25     !5 号线
LSTR,25,6     !6 号线
LSTR,7,8      !7 号线
LSTR,8,9      !8 号线
LSTR,9,10     !9 号线
LSTR,10,11    !10 号线
LSTR,11,26    !11 号线
LSTR,26,12    !12 号线
LSTR,13,14    !13 号线
LSTR,14,15    !14 号线
LSTR,15,16    !15 号线
LSTR,16,17    !16 号线
LSTR,17,27    !17 号线
LSTR,27,18    !18 号线
LSTR,19,20    !19 号线
LSTR,20,21    !20 号线
LSTR,21,22    !21 号线
LSTR,22,23    !22 号线
LSTR,23,28    !23 号线
LSTR,28,24    !24 号线
!**第一层外围横撑
LSTR,1,13     !25 号线
LSTR,13,19    !26 号线
LSTR,19,7     !27 号线
LSTR,7,1      !28 号线
!**第二层外围横撑
LSTR,2,14     !29 号线
LSTR,14,20    !30 号线
LSTR,20,8     !31 号线
LSTR,8,2      !32 号线
!**第三层外围横撑
```

LSTR,3,15	!33 号线
LSTR,15,21	!34 号线
LSTR,21,9	!35 号线
LSTR,9,3	!36 号线
!**第四层外围横撑	
LSTR,4,16	!37 号线
LSTR,16,22	!38 号线
LSTR,22,10	!39 号线
LSTR,10,4	!40 号线
!**第五层外围横撑	
LSTR,5,17	!41 号线
LSTR,17,23	!42 号线
LSTR,23,11	!43 号线
LSTR,11,5	!44 号线
!**第六层外围横撑	
LSTR,25,27	!45 号线
LSTR,27,28	!46 号线
LSTR,28,26	!47 号线
LSTR,26,25	!48 号线
!**第一层内部横撑	
LSTR,1,29	!49 号线
LSTR,13,29	!50 号线
LSTR,19,29	!51 号线
LSTR,7,29	!52 号线
!**第二层内部横撑	
LSTR,2,30	!53 号线
LSTR,14,30	!54 号线
LSTR,20,30	!55 号线
LSTR,8,30	!56 号线
!**第三层内部横撑	
LSTR,3,31	!57 号线
LSTR,15,31	!58 号线
LSTR,21,31	!59 号线
LSTR,9,31	!60 号线
!**第四层内部横撑	
LSTR,4,32	!61 号线
LSTR,16,32	!62 号线
LSTR,22,32	!63 号线
LSTR,10,32	!64 号线
!**第五层内部横撑	
LSTR,5,33	!65 号线
LSTR,17,33	!66 号线
LSTR,23,33	!67 号线
LSTR,11,33	!68 号线

## ! \*\*第六层内部横撑

LSTR,25,35 !69 号线

LSTR,27,35 !70 号线

LSTR,28,35 !71 号线

LSTR,26,35 !72 号线

## ! \*\*外围斜撑

LSTR,1,8 !73 号线

LSTR,8,3 !74 号线

LSTR,3,10 !75 号线

LSTR,10,5 !76 号线

LSTR,5,26 !77 号线

LSTR,19,14 !78 号线

LSTR,14,21 !79 号线

LSTR,21,16 !80 号线

LSTR,16,23 !81 号线

LSTR,23,27 !82 号线

LSTR,7,20 !83 号线

LSTR,20,9 !84 号线

LSTR,9,22 !85 号线

LSTR,22,11 !86 号线

LSTR,11,28 !87 号线

LSTR,13,2 !88 号线

LSTR,2,15 !89 号线

LSTR,15,4 !90 号线

LSTR,4,17 !91 号线

LSTR,17,25 !92 号线

## ! \*\*甲板主导管

LSTR,1,36 !93 号线

LSTR,7,37 !94 号线

LSTR,19,38 !95 号线

LSTR,13,39 !96 号线

LSTR,71,36 !97 号线

LSTR,72,37 !98 号线

LSTR,73,38 !99 号线

LSTR,74,39 !100 号线

## ! \*\*水面以上外围支撑

LSTR,1,37 !101 号线

LSTR,37,71 !102 号线

LSTR,19,39 !103 号线

LSTR,39,73 !104 号线

LSTR,13,36 !105 号线

LSTR,36,74 !106 号线

LSTR,7,38 !107 号线

LSTR,38,72 !108 号线

## ! \*\*梁结构

LSTR,36,68	!109 号线
LSTR,68,37	!110 号线
LSTR,37,69	!111 号线
LSTR,69,38	!112 号线
LSTR,38,70	!113 号线
LSTR,70,39	!114 号线
LSTR,39,67	!115 号线
LSTR,67,36	!116 号线
LSTR,52,67	!117 号线
LSTR,52,68	!118 号线
LSTR,52,69	!119 号线
LSTR,52,70	!120 号线
LSTR,40,44	!121 号线
LSTR,44,53	!122 号线
LSTR,53,45	!123 号线
LSTR,45,41	!124 号线
LSTR,41,54	!125 号线
LSTR,54,46	!126 号线
LSTR,46,55	!127 号线
LSTR,55,47	!128 号线
LSTR,47,56	!129 号线
LSTR,56,42	!130 号线
LSTR,42,48	!131 号线
LSTR,48,57	!132 号线
LSTR,57,49	!133 号线
LSTR,49,43	!134 号线
LSTR,43,58	!135 号线
LSTR,58,50	!136 号线
LSTR,50,59	!137 号线
LSTR,59,51	!138 号线
LSTR,51,60	!139 号线
LSTR,60,40	!140 号线
LSTR,59,67	!141 号线
LSTR,69,55	!142 号线
LSTR,58,66	!143 号线
LSTR,66,65	!144 号线
LSTR,65,64	!145 号线
LSTR,64,56	!146 号线
LSTR,50,39	!147 号线
LSTR,38,47	!148 号线
LSTR,51,36	!149 号线
LSTR,37,46	!150 号线
LSTR,60,61	!151 号线

LSTR,61,62	!152 号线
LSTR,62,63	!153 号线
LSTR,63,54	!154 号线
LSTR,49,66	!155 号线
LSTR,66,39	!156 号线
LSTR,36,61	!157 号线
LSTR,61,44	!158 号线
LSTR,57,65	!159 号线
LSTR,65,70	!160 号线
LSTR,68,62	!161 号线
LSTR,62,53	!162 号线
LSTR,48,64	!163 号线
LSTR,64,38	!164 号线
LSTR,37,63	!165 号线
LSTR,63,45	!166 号线
LSTR,78,93	!167 号线
LSTR,93,85	!168 号线
LSTR,85,94	!169 号线
LSTR,94,86	!170 号线
LSTR,86,95	!171 号线
LSTR,95,75	!172 号线
LSTR,84,101	!173 号线
LSTR,101,74	!174 号线
LSTR,74,102	!175 号线
LSTR,102,71	!176 号线
LSTR,71,96	!177 号线
LSTR,96,79	!178 号线
LSTR,92,100	!179 号线
LSTR,100,105	!180 号线
LSTR,105,87	!181 号线
LSTR,87,103	!182 号线
LSTR,103,97	!183 号线
LSTR,97,88	!184 线
LSTR,83,99	!185 号线
LSTR,99,73	!186 号线
LSTR,73,104	!187 号线
LSTR,104,72	!188 号线
LSTR,72,98	!189 号线
LSTR,98,80	!190 号线
LSTR,77,91	!191 号线
LSTR,91,82	!192 号线
LSTR,82,90	!193 号线
LSTR,90,81	!194 号线
LSTR,81,89	!195 号线

LSTR,89,76	!196 号线
LSTR,78,84	!197 号线
LSTR,84,92	!198 号线
LSTR,92,83	!199 号线
LSTR,83,77	!200 号线
LSTR,93,101	!201 号线
LSTR,101,100	!202 号线
LSTR,100,99	!203 号线
LSTR,99,91	!204 号线
LSTR,85,74	!205 号线
LSTR,74,105	!206 号线
LSTR,105,73	!207 号线
LSTR,73,82	!208 号线
LSTR,94,102	!209 号线
LSTR,102,87	!210 号线
LSTR,87,104	!211 号线
LSTR,104,90	!212 号线
LSTR,86,71	!213 号线
LSTR,71,103	!214 号线
LSTR,103,72	!215 号线
LSTR,72,81	!216 号线
LSTR,95,96	!217 号线
LSTR,96,97	!218 号线
LSTR,97,98	!219 号线
LSTR,98,89	!220 号线
LSTR,75,79	!221 号线
LSTR,79,88	!222 号线
LSTR,88,80	!223 号线
LSTR,80,76	!224 号线
LSTR,52,87	!225 号线
LSTR,71,44	!226 号线
LSTR,72,80	!227 号线
LSTR,73,48	!228 号线
LSTR,74,49	!229 号线
LSTR,65,100	!230 号线
LSTR,97,62	!231 号线
LSTR,45,72	!232 号线
LSTR,36,75	!233 号线
LSTR,37,76	!234 号线
LSTR,38,77	!235 号线
LSTR,39,78	!236 号线
LSTR,57,92	!237 号线
LSTR,53,88	!238 号线
!**一层甲板结构平面	

A,43,49,66,58  
A,58,66,39,50  
A,50,39,67,59  
A,59,67,36,51  
A,51,36,61,60  
A,60,61,44,40  
A,49,57,65,66  
A,66,65,70,39  
A,39,70,52,67  
A,67,52,68,36  
A,36,68,62,61  
A,61,62,53,44  
A,57,48,64,65  
A,65,64,38,70  
A,70,38,69,52  
A,52,69,37,68  
A,68,37,63,62  
A,62,63,45,53  
A,48,42,56,64  
A,64,56,47,38  
A,38,47,55,69  
A,69,55,46,37  
A,37,46,54,63  
A,63,54,41,45  
! \*\* 二层甲板结构平面  
A,78,84,101,93  
A,93,101,74,85  
A,85,74,102,94  
A,94,102,71,86  
A,86,71,96,95  
A,95,96,79,75  
A,84,92,100,101  
A,101,100,105,74  
A,74,105,87,102  
A,102,87,103,71  
A,71,103,97,96  
A,96,97,88,79  
A,92,83,99,100  
A,100,99,73,105  
A,105,73,104,87  
A,87,104,72,103  
A,103,72,98,97  
A,97,98,80,88  
A,83,77,91,99

```

A,99,91,82,73
A,73,82,90,104
A,104,90,81,72
A,72,81,89,98
A,98,89,76,80
!*****有限元模型网格划分
!**水面以下泥线以上主导管
LSEL,S,,,1,5      !从全部对象中选中 1~5 号线
LSEL,A,,,7,11     !继续选中 7~11 号
LSEL,A,,,13,17    !继续选中 13~17 号
LSEL,A,,,19,23    !继续选中 19~23 号
LATT,1,2,2        !为已选中的线分配材料模型、实常数、单元类型
LESIZE,ALL,,,6    !将已选中的线每条划分为 5 份
LMESH,ALL         !线单元划分
!**泥线以下主导管
LSEL,S,,,6        !从全部对象中选中 6 号线
LSEL,A,,,12       !继续选中 12 号
LSEL,A,,,18       !继续选中 18 号
LSEL,A,,,24       !继续选中 24 号
LATT,1,1,1        !为已选中的线分配材料模型、实常数、单元类型
LESIZE,ALL,,,2    !将已选中的线每条划分为 2 份
LMESH,ALL         !线单元划分
!**层外围横支撑与水面以上主导管 PIPE59
LSEL,S,,,25,48    !从全部对象中选中 25~48 号线
LSEL,A,,,93,100
LATT,1,3,2        !为已选中的线分配材料模型、实常数、单元类型
LESIZE,ALL,,,6    !将已选中的线每条划分为 4 份
LMESH,ALL         !线单元划分
!**内围横支撑与外围斜向支撑 PIPE59
LSEL,S,,,49,92    !从全部对象中选中 49~72 号线
LSEL,A,,,101,108  !继续选中 101~108 号线
LSEL,A,,,226,238
LATT,1,4,2        !为已选中的线分配材料模型、实常数、单元类型
LESIZE,ALL,,,8    !将已选中的线每条划分为 5 份
LMESH,ALL         !线单元划分
!**梁结构
LSEL,S,,,109,225  !从全部对象中选中 109~232 号线
LATT,1,5,3,,,1    !为已选中的线分配材料模型、实常数、单元类型
LESIZE,ALL,,,6    !将已选中的线每条划分为 6 份
LMESH,ALL         !线单元划分
!**甲板平面结构
ASEL,A,,,ALL      !选中所有面
AATT,1,6,4        !为已选面分配材料模型、实常数、单元类型
AMESH,ALL         !面划分

```



```

/ESHAPE,1.0          !打开模型形状控制开关
NUMMRG,NODE,, ,LOW  !合并重合节点
NUMCMP,ALL          !压缩编号
FINISH
!****波流耦合值搜索
*DIM,WLX,ARRAY,361,1,1, , , !定义用于储存相位角的一维数组
*DIM,WLFX,ARRAY,361,1,1, , , !定义用于储存波流耦合力值的一维数组
*DO,I,0,360,1        !定义相位角范围 0°~360°, 并开始循环
*SET,WLX(I+1),I      !将相位角值储存于 WLX 数组
/PREP7              !进入前处理模块
NSEL,S,LOC,Z,-50
CP,,ALL,ALL
ALLSEL
D,125,ALL
ACEL,0,0,9.8
FINISH
TBDATA,81,I         !将相位角取值赋予 Water Table
/SOLU               !进入求解计算模块
TIME,1              !定义结束结束时间
SOLVE               !开始求解
*GET,RFX,NODE,125,RF,FX !提取该节点 X 方向的反力
*SET,WLFX(I+1),-(RFX) !将所提取的反力取相反数后储存于 WLFX 数组
*ENDDO              !结束循环
*CREATE,ANSUO        !创建一个文件
*CFOPEN,'RFORCE','OUT'," !打开文件
*VWRITE,WLX(1),WLFX(1), , , , , , !写入文件
(3F15.4)              !
*CFCLOS
*END
/INPUT,ANSUO
!*****静力计算过程
/SOLU               !进入求解计算模块
ANTYPE,0            !设置求解类型为结构静力分析
NSEL,S,LOC,Z,-50    !选中 Z=-50 处的节点
D,ALL,ALL           !约束所选节点全部自由度
ALLSEL              !选中所有对象
ACEL,0,0,9.8        !施加重力载荷
SOLVE               !执行求解

```

```

SAVE                                !保存求解结果
FINISH                              !退出求解计算模块
!*****静力分析后处理过程
/POST1                              !进入通用后处理器
SET, LAST                           !读入计算结果文件
/ESHAPE, 1.0                        !打开模型形状开关
PLDISP, 0                           !绘制结构整体变形图
PLNSOL, U, SUM, 0, 1.0              !绘制结构位移等值图
PLNSOL, S, EQV, 0, 1.0              !绘制结构等效应力分布云图
!*****模态分析计算
/SOLU                                !进入求解模块
ANTYPE, MODAL                       !设置分析类型为模态分析
MODEOPT, LANB, 6                    !模态分析选项
MXPAND, 6                           !设置扩展 6 阶模态
NSEL, S, LOC, Z, -50                !选中 Z=-50 处的节点
D, ALL, ALL                         !约束所选节点全部自由度
ALLSEL                              !选中所有对象
ACEL, 0, 0, 9.8                     !设置重力加速度
SOLVE                               !开始模态求解
SAVE                                !保存求解结果
FINISH                              !退出求解模块
!*****模态分析后处理
/POST1                              !进入通用后处理器
ESHAPE, 1.0                        !打开结构形状开关
SET, LIST                           !列表显示结构自振频率计算结果
SET, FIRST                          !读入一阶模态
PLDISP, 0                           !显示结构一阶模态
SET, NEXT                           !读入二阶模态
PLDISP, 0                           !显示结构二阶模态
SET, NEXT                           !读入三阶模态
PLDISP, 0                           !显示结构三阶模态
SET, NEXT                           !读入四阶模态
PLDISP, 0                           !显示结构四阶模态
FINISH                              !退出后处理模块
!****波浪载荷作用下瞬态动力分析
/PREP7                              !进入前处理模块

```

```

TB,81,0          !修改 WATERTABLE 里的波浪相位角为 0°
/SOLU            !进入求解计算模块
ANTYPE,TRANS     !设置求解类型为瞬态动力分析
ALPHAD,0.1575    !设置阻尼系数  $\alpha$ 
BETAD,2.5389E-03 !设置阻尼系数  $\beta$ 
TRNOPT,FULL      !设定瞬态动力求解方法
NLGEOM,ON        !打开大变形开关
NSUBST,1,,,1
NSEL,S,LOC,Z,-50 !选中 Z=-50 处的节点
D,ALL,ALL        !约束所选节点全部自由度
ALLSEL           !选中所有对象
ACEL,0,0,9.8
*DO,I,10E-12,100,0.2 !设定结束时间及子步时间开始循环计算
TIME,I           !循环计算结束时间
OUTPES,ALL,ALL   !输出结果设置
SOLVE,SAVE       !执行求解并保存
*ENDDO           !结束循环求解计算
!*****时程后处理过程
/POST26          !进入时程后处理器 POST26
NSOL,2,1173,U,X,Deflection !定义 1 号节点处的 X 方向位移为 2 号时程变量
/AXLAB,X,TIME    !定义 X 轴标题
/AXLAB,Y,UX      !定义 Y 轴标题
PLVAR,2          !绘制时程位移曲线
ESOL,3,125,,SMIS,15,STRESS !定义 17 号单元的等效应力为 3 号时程变量
/AXLAB,X,TIME    !定义 X 轴标题
/AXLAB,Y,Stress  !定义 Y 轴标题
PRVAR,3          !绘制时程应力曲线
FINISH           !退出 POST26

```

### 主要参考文献:

1. 龚曙光 主编 《ANSYS 工程应用实例解析》机械工业出版社 2003
2. 博嘉科技 编著 《有限元分析软件—ANSYS 融会与贯通》中国水利水电出版社 2002
3. 尚晓峰 等 编著 《ANSYS 结构有限元高级分析方法与范例应用》中国水利水电出版社 2006
4. 李路 海洋石油工业的发展 科技创新导报 2007
5. 任克忍等 海洋石油水下装备现状及发展趋势 石油机械 2008
6. 李茜 杨树耕 采用 ANSYS 程序的自升式平台结构有限元动力分析 中国海洋平台 2003
7. 杨进等 风浪流作用下隔水导管强度及安全性计算 中国海上油气 2006
8. 廖彦波 杨进 双圆夹层钢管混凝土在钻井隔水导管中的应用 钢结构 2008
9. 张建勇 浅海独桩平台有限元分析及合理结构型式研究 硕士论文 2005
10. 韩志强. 海洋平台桩基计算与施工方法探讨. 中国海洋平台. 2002. 12
11. 刘庭权等. 基于桩--土--结构相互作用的海洋平台结构抗震优化设计. 船舶力学. 2004. 8
12. 卢成原. 考虑群桩效应的海洋平台桩--土结构共同工作研究. 海洋工程. 1996. 2
13. 黄熠 萧林 李炎军 单筒三井技术在涠洲 6-1 油田的应用 石油钻采工艺 2007